Q:

RegionPlot3D is giving wrong plot

I am seeing discrepancies between plots obtained for same region using Region and RegionPlot3D,
When I use 
reg4 = ImplicitRegion[
   A1 > 0 && 
    A2 > 0 && (\[Mu] < -((3 Abs[A1])/2)) && (\[Mu] < -(3/2) Sqrt[3]
        Abs[A2]) && (2 \[Mu] + Sqrt[
       A1^2 + 16 A2^2 + 8 Abs[A1] Abs[A2]] < 0), {{A1, 0, 2}, {A2, 0, 
     2}, {\[Mu], -1.5, 0}}];
Region[reg4, Axes -> True, AxesLabel -> {"A1", "A2", "\[Mu]"}]

If I use RegionPlot3D then I get
RegionPlot3D[
 A1 > 0 && 
  A2 > 0 && (\[Mu] < -((3 Abs[A1])/2)) && (\[Mu] < -(3/2) Sqrt[3]
      Abs[A2]) && (2 \[Mu] + Sqrt[
     A1^2 + 16 A2^2 + 8 Abs[A1] Abs[A2]] < 0), {A1, 0, 2}, {A2, 0, 
  2}, {\[Mu], -1.5, 0}, Axes -> True, 
 AxesLabel -> {"A1", "A2", "\[Mu]"}]

As if the surface is smoothened in this figure, so the plot is also wrong.  
Why two different graphics for the same region? 

A:

Clear["Global`*"]

The Abs appear to be causing most of the roughness of the Region plot. Perhaps because Abs does not have a derivative. Since you are dealing with real numbers, Abs[x] == Sqrt[x^2].
reg4 = ImplicitRegion[
    A1 > 0 && 
     A2 > 0 && (μ < -((3 Abs[A1])/2)) &&
     (μ < -(3/2) Sqrt[3] Abs[
         A2]) &&
     (2 μ + Sqrt[A1^2 + 16 A2^2 + 8 Abs[A1] Abs[A2]] < 0),
    {{A1, 0, 2}, {A2, 0, 2}, {μ, -1.5, 0}}] /.
   Abs[x_] :> Sqrt[x^2];

EDIT: Alternatively, use FullSimplify within definition of reg4
reg4 = ImplicitRegion[
   A1 > 0 && 
     A2 > 0 &&
     (μ < -((3 Abs[A1])/2)) &&
     (μ < -(3/2) Sqrt[
         3] Abs[A2]) &&
     (2 μ + 
        Sqrt[A1^2 + 16 A2^2 + 8 Abs[A1] Abs[A2]] < 0) //
    FullSimplify,
   {{A1, 0, 2}, {A2, 0, 2}, {μ, -3/2, 0}}];

reg = Region[reg4,
  Axes -> True,
  AxesLabel -> {"A1", "A2", "μ"},
  PerformanceGoal -> "Quality"]

To get a more accurate RegionPlot3 increase the PlotPoints and MaxRecursion
RegionPlot3D[
 A1 > 0 && A2 > 
   0 && (μ < -((3 Abs[A1])/2)) && (μ < -(3/2) Sqrt[3] Abs[
      A2]) && (2 μ + Sqrt[A1^2 + 16 A2^2 + 8 Abs[A1] Abs[A2]] < 0), {A1, 
  0, 2}, {A2, 0, 2}, {μ, -1.5, 0}, Axes -> True, 
 AxesLabel -> {"A1", "A2", "μ"}, PlotPoints -> 100, MaxRecursion -> 5]

